
Figure S1. (a) Target mathematical model for the initial dQTcF linear mixed-effects model, and 
(b) formula (fit in R) describing the final dQTcF model components. 

(a)

∆QTcFijk = (θ0+η0,i ) + θ1 ∙ TRTj + ∑ (θ(2,l)+η2,il) ∙ Cijkl + θ3 ∙ TIMEj + θ4 ∙ (QTcFi1 j=0 – QTcF0)

∆QTcFijk, change from baseline in dQTcF for patient i receiving treatment j at time k; θ0, population mean intercept in 
the absence of a treatment effect; η0,i, random effect associated with the intercept θ0; θ1, fixed effect associated with the 
treatment TRTj (0=placebo, 1=active drug); θ2,l, population mean slope of the assumed linear association between the 
concentration of analyte l (molibresib, active metabolite composite [GSK3529246] or TAM) and ∆QTcijk; η2,il, random effect 
associated with the slope θ2,l; Cijkl, concentration of the analyte l for patient i receiving treatment j at time k; θ3, fixed effect 
associated with time; θ4, fixed effect associated with baseline QTci,j=0; QTcF0, overall mean of QTci,j=0 across patients,  
i.e., the mean of all the baseline dQTcF values.

(b)

DQTCF ~ INT + DM ∙ DMQTCF + GSK3529246(P145∙(OCC==1,4,5)+P2∙(OCC==2)+P3∙(OCC==3))

DQTCF, QT interval prolongation corrected by Fridericia formula and by baseline; INT, model intercept; DM, parameter 
describing the effect of DMQTCF; DMQTCF, the individual specific difference of baseline QTcF from population mean;  
Px, occasion-specific exponents of the molibresib active metabolite composite (GSK3529246) effect for occasion x.

n=1

l _____

Figure S6. Heart rate change from baseline with molibresib and GSK3529246  

Dots represent observations; solid purple line represents the linear model fit; dashed purple line represents smooth 
through the data; horizontal grey lines indicate 0 and 10 bpm.

bpm, beats per minute; PK, pharmacokinetics.

Figure S8. Incidence of nausea, dysgeusia, diarrhea, and vomiting versus TAM  
exposure metrics.

Dots represent individual data points (1.0: nausea/dysgeusia/diarrhea/vomiting Grade 2 or greater, 0.0: No nausea/
dysgeusia/diarrhea/vomiting or Grade 1 nausea/dysgeusia/diarrhea/vomiting); solid lines represent the model fits; shaded 
green areas represent 95% confidence intervals.

AUC, area under the concentration–time curve; Cmax, maximum plasma concentration; Cmin, minimum plasma 
concentration; TAM, total active moiety.



Supplementary Text 1. Final PK/PD Platelet model code.

;; 1. Based on: 231

;; 2. Description:

;;    Final Platelet model

;; 3. Label:

;;    Friberg model

;; 4. Structural model:

;;    Friberg model

;; 5. Covariate model:

;;    Forward step 1

;; 6. Inter-individual variability:

;;    Baseline

;; 7. Inter-occasion variability:

;;    NONE

;; 8. Residual variability:

;;    Proportional 

;; 9. Estimation:

;;    FOCE+I

$SIZES      PD=-100

$PROBLEM    GSK BET

$INPUT      STUDY ID SUBJID PARTN AMTMG AMTUG AMT II ADDL LINDV DV

            BPLT DAY WEEK TIME RTLD ADOSE EVID MDV OCC DOSENUM QDBID

            AGE SEX WT HT BBSA BBMI RACEN ETHN CANCER BECOG ECOG

            EXFLAG FDARACEN FOOD PPTF PTAX BAST AST HGB BHGB IALAG1

            ICL IV IKA IV2 IQ IKIN ISLP NIMCL IMV IMKA IMV2 IMQ IMCL

$DATA      ../nm_pkpd_data_25SEP19_param32.csv IGNORE=@



            IGNORE=(EXFLAG.GT.0)

; --------------------------------------------------------------------------------

$SUBROUTINE ADVAN13 TOL=12

$MODEL      COMP=(ABS,DEFDOSE) COMP=(LIVER) COMP=(METABS)

            COMP=(CENT,DEFOBS) COMP=(PERIPH) COMP=(IND) COMP=(METCENT)

            COMP=(METPERIPH) COMP=(PROLIFERATING) COMP=(TRANSIT)

            COMP=(TRANSIT) COMP=(TRANSIT) COMP=(CIRCULATING)

; --------------------------------------------------------------------------------

$PK 

; Stratification by dose levels for VPC

STRT=1 ; Doses below 50 mg

IF(ADOSE.GE.50) STRT=2   ; Doses 50-60 mg

IF(ADOSE.GT.60) STRT=3   ; Doses 65-75 mg

IF(ADOSE.GT.75) STRT=4   ; Doses 80-100 mg

; --------------    PLATELETS                 ------------------------------------

COVCAN_FB  = 1

IF(CANCER.EQ.3) COVCAN_FB = (1+THETA(6))

IF(CANCER.GE.7) COVCAN_FB = (1+THETA(7))

TVBASE    = THETA(1)

TVKTRANS  = THETA(2)

TVFBPOWER = THETA(3)



TVEC50    = THETA(4)

TVEMAX    = THETA(5)

MU_1 = TVBASE

MU_2 = TVKTRANS

MU_3 = TVFBPOWER

MU_4 = TVEC50

MU_5 = TVEMAX

BASE    = EXP(MU_1 + ETA(1))

KTRANS  = EXP(MU_2 + ETA(2))

FBPOWER = COVCAN_FB*EXP(MU_3 + ETA(3))

EC50    = EXP(MU_4 + ETA(4))

EMAX    = EXP(MU_5 + ETA(5))

A_0(9)  = BASE

A_0(10) = BASE

A_0(11) = BASE

A_0(12) = BASE

A_0(13) = BASE

; --------------         PK from run 32        ------------------------------------

; Parent

KA    = IKA

CL    = ICL

V     = IV



QH    = EXP(4.007333)

VH    = EXP(0.4054651)

Q     = IQ

V2    = IV2

ALAG1 = IALAG1

SLP   = ISLP

KIN   = IKIN

; Metabolite

POWAST=-0.194

COVAST_MCL = (AST/28)**POWAST

MKA    = IMKA

MCL    = IMCL*COVAST_MCL

MV     = IMV

MQ     = IMQ

MV2    = IMV2

; ---- initialize enzyme CMT ---------------

ENZBASE   = 1

KOUT   = KIN/ENZBASE

A_0(6) = ENZBASE

; Rate Constants

K42 = QH/V

K45 = Q/V

K54 = Q/V2



K70 = MCL/MV

K78 = MQ/MV

K87 = MQ/MV2

IF(NEWIND.LT.2) THEN

 IFL = 0 

 TAD = 0.0 

ENDIF

 

IF(EVID.EQ.1) THEN

 TDOS = TIME 

 TAD = 0.0 

 IFL = 1 

ENDIF 

IF(IFL.EQ.1.AND.EVID.NE.1) TAD = TIME - TDOS

; --------------------------------------------------------------------------------

$DES

; Parent

CH = A(2)/VH + 1E-9 ; liver concentrations

IND = SLP*CH  ; linear induction of liver enzymes

EH = CL*A(6)/QH  ; heaptic extraction ratio

FH = 1 - EH   ; fraction escaping first-pass metabolism



K24 = QH*FH/VH

K23 = QH*EH/VH

DADT(1) = -KA*A(1)

DADT(2) = KA * A(1) - K24 * A(2) + K42 * A(4) - K23 * A(2)

DADT(4) = K24 * A(2) - K42 * A(4) - K45*A(4) + K54*A(5)

DADT(5) = K45*A(4) - K54*A(5)

DADT(6) = KIN*(1 + IND) - KOUT*A(6)

A6 = A(6)

; Metabolite

DADT(3) = K23*A(2) - MKA*A(3)

DADT(7) = MKA*A(3) - K78*A(7) + K87*A(8) - K70*A(7)

DADT(8) = K78*A(7) - K87*A(8)

; Platelets

EFFCONC=(A(4)/V + A(7)/MV)

DEFF =  (EMAX * EFFCONC/(EC50 + EFFCONC))

RATIO    = BASE/A(13)

FEEDBACK = RATIO**FBPOWER

DADT(9)  = KTRANS*A(9)*(1-DEFF)*FEEDBACK - KTRANS*A(9) ;prol cells

DADT(10) = KTRANS*A(9)  - KTRANS*A(10) ;transit 1

DADT(11) = KTRANS*A(10) - KTRANS*A(11) ;transit 2

DADT(12) = KTRANS*A(11) - KTRANS*A(12) ;transit 3

DADT(13) = KTRANS*A(12) - KTRANS*A(13)  ;circ plat



; --------------------------------------------------------------------------------

$ERROR

IPRED = LOG(A(13)+0.0000001)

Y = IPRED + EPS(1)

; --------------------------------------------------------------------------------

$THETA  (2.5,5.55,7.5) ; Baseline Platelet Count; (10^9/L); LOG

$THETA  (-5,-3.36,-1) ; Platelet Transit Rate; h-1; LOG

$THETA  (-3,-1.94,0) ; Feedback Power; -; LOG

$THETA  (-15,1.57) ; EC50; nM; LOG

$THETA  -0.001 FIX ; EMAX; -; LOG

$THETA  (-0.99,2) ; FEEDBACK~CANCER3; ; LIN

$THETA  (-0.99,2) ; FEEDBACK~CANCER78; ; LIN

; --------------------------------------------------------------------------------

$OMEGA  0.125  ; Baseline Platelet Count_

$OMEGA  0.0167  ; Platelet Transit Rate_

$OMEGA  0.256  ; Feedback Power_

$OMEGA  0.363  ;      EC50_

$OMEGA  0  FIX  ;      EMAX_

; --------------------------------------------------------------------------------

$SIGMA  0.0765  ; Proportional Error

; --------------------------------------------------------------------------------

$ESTIMATION METHOD=COND MAXEVAL=9999 INTER SIGL=9 NSIG=3 NOABORT FAST

            PRINT=1



$COVARIANCE PRINT=E UNCONDITIONAL SIGL=9

$TABLE      ID TAD DV EVID TIME RTLD IPRED CWRES BASE KTRANS FBPOWER

            EC50 EMAX ETAs(1:LAST) BPLT DAY WEEK ADOSE OCC DOSENUM

            QDBID AGE SEX WT HT BBSA BBMI RACEN ETHN CANCER BECOG

            EXFLAG FDARACEN FOOD NOPRINT ONEHEADER FORMAT=s1PE13.7

            FILE=sdtab219

$TABLE      ID BASE KTRANS FBPOWER EC50 EMAX ETAs(1:LAST) NOPRINT

            ONEHEADER FORMAT=s1PE13.7 FILE=patab219

$TABLE      ID BPLT ADOSE AGE WT HT BBMI BBSA HGB BHGB NOPRINT

            ONEHEADER FORMAT=s1PE13.7 FILE=cotab219

$TABLE      ID PARTN SEX RACEN ETHN CANCER ECOG BECOG PPTF PTAX

            NOPRINT ONEHEADER FORMAT=s1PE13.7 FILE=catab219


